Wiederholung: Sheets 1-4

Wiederhole die Themen: Euklidische Vektorräume, Fünfzackige Sterne: Anschauliche Anwendung von Eigenwerten, Normierte Räume: Erste Beispiele, Konvergente Folgen in normierten Räumen, Spektralsatz, Topologie, Direkte Zerlegungen und Endomorphismen, Differentiation: Definition, Eigenschaften und Beispiele, Monotonie, Extrema, Konvexität, Funktionenräume, die unter Differentiation abgeschlossen sind!
Wir stellen Fragen im Testat!

Mittelwertsätze

Aufgaben: 3

```
> restart;
```

Mittelwertsätze

Die Mittelwertsätze (Satz von Rolle, Mittelwertsatz, verallgemeinerter Mittelwertsatz) sind wertvolle Hilfsmittel bei Funktionsuntersuchungen, insbesondere auch bei Grenzwertbestimmungen.

MATH: Sei
$$f\colon [a, b] \to \mathbb{R}$$
stetig auf $[a, b]$ und differenzierbar auf (a, b).
1) (Satz von Rolle) Gilt $f(a) = f(b) = 0$, so existiert ein $x \in (a, b)$ mit $f'(x) = 0$.
2) (Mittelwertsatz der Differentialrechnung) Im Allgemeinen existiert ein $x \in (a, b)$ mit
$$f'(x) = \frac{f(b) - f(a)}{b - a}$$
3) (Verallgemeinerter Mittelwertsatz) Ist $g\colon [a, b] \to \mathbb{R}$ ebenfalls stetig auf $[a, b]$ und differenzierbar auf (a, b), so existiert ein $x \in (a, b)$ mit
$$g'(x) \cdot (f(b) - f(a)) = f'(x) \cdot (g(b) - g(a)).$$

DENKANSTOSS: Führe 2) auf 1) zurück und 3) auf 2).

ÜBUNG [01]:
Gib mit Hilfe von 2) alle differenzierbaren Funktionen $f\colon \mathbb{R} - \{0\} \to \mathbb{R}$ an, deren Ableitung identisch gleich Null ist. Hinweis: Beachte die Definitionslücke, der Mittelwertsatz gilt nur auf einem abgeschlossenen Intervall!

DENKANSTOSS: Eine Schranke für $f'(x)$ auf (a, b) führt zu Schranken für $f(x) - f(a)$ auf $[a, b]$.

Regel von L'Hospital

MATH: Eine wichtige Anwendung des verallgemeinerten Mittelwertsatzes ist die Regel von L'Hospital in ihren verschiedensten Versionen, die beispielsweise den Grenzwert von Quotienten zweier Funktionen, die gegen Null konvergieren, auf den Grenzwert des Quotienten der beiden Ableitungen reduziert:
Seien $a, b \in \mathbb{R}, a < b$ oder $b = \infty$. Gegeben seien zwei differenzierbare Funktionen $f: (a, b) \rightarrow \mathbb{R}$ und $g: (a, b) \rightarrow \mathbb{R}$ mit $g'(x) \neq 0$ für alle $x \in (a, b)$. Gilt

$$\lim_{x \to b, x \le b} f(x) = \lim_{x \to b, x \le b} g(x) = 0 \quad \text{oder}$$

$$\lim_{x \to b, x \le b} f(x), \lim_{x \to b, x \le b} g(x) \in \{-\infty, \infty\}$$

und existiert

$$\lim_{x \to b, x \le b} \frac{f'(x)}{g'(x)} = L \in \mathbb{R} \cup \{\pm\infty\}$$

so gilt

$$\lim_{x \to b, x \le b} \frac{f(x)}{g(x)} = L.$$

.

BEISPIEL:

```
> a:=x-1/6*x^3+1/120*x^5-1/5040*x^7+1/362880*x^9;
```

$$a := x - \frac{1}{6} x^3 + \frac{1}{120} x^5 - \frac{1}{5040} x^7 + \frac{1}{362880} x^9 \quad \textbf{(2.2.1)}$$

```
> limit(sin(x)-a,x=0);
```

$$0 \quad \textbf{(2.2.2)}$$

```
> limit(x^11,x=0);
```

$$0 \quad \textbf{(2.2.3)}$$

Gesucht ist

```
> limit((sin(x)-a)/x^11,x=0);
```

$$-\frac{1}{39916800} \quad \textbf{(2.2.4)}$$

Da die ersten 10 Ableitungen von x^{11} noch alle an der Stelle Null den Wert Null haben, müssen wir wahrscheinlich 11 mal ableiten:

```
> map(i->limit(diff(sin(x)-a,x$i),x=0),[$1..11]);
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -1] \quad \textbf{(2.2.5)}$$

```
> diff(x^11,x$11);
```

$$39916800 \quad \textbf{(2.2.6)}$$

Womit die MAPLE-Bestimmung des Grenzwertes verifiziert ist.

ÜBUNG [02]:

Prüfe Konvergenz und bestimme gegebenenfalls den Grenzwert mit Hilfe der L'Hospitalschen Regel von

1) $\frac{g}{h}$

2) $\frac{g^2}{h^2}$

3) $\frac{g}{h^2}$

für x gegen 0 mit

```
> g := ln(1+x)-x+1/2*x^2-1/3*x^3+1/4*x^4-1/5*x^5+1/6*x^6-1/7*x^7+1/8*x^8-1/9*x^9+1/10*x^10;
```

$$g := \ln(1+x) - x + \frac{1}{2}x^2 - \frac{1}{3}x^3 + \frac{1}{4}x^4 - \frac{1}{5}x^5 + \frac{1}{6}x^6 - \frac{1}{7}x^7 + \frac{1}{8}x^8 - \frac{1}{9}x^9 + \frac{1}{10}x^{10}$$

```
h:=arctan(x)-x+1/3*x^3-1/5*x^5+1/7*x^7-1/9*x^9;
```

$$h := \arctan(x) - x + \frac{1}{3}x^3 - \frac{1}{5}x^5 + \frac{1}{7}x^7 - \frac{1}{9}x^9 \tag{2.2.8}$$

MATH: Wichtig ist die Einsicht, daß $x \mapsto x$ schneller nach unendlich strebt als jede Potenz des Logarithmus, was man leicht mit einer geeigneten Variante von L'Hospital sieht:

ÜBUNG [03]:

Zeige per Induktion mit Hilfe der L'Hospitalschen Regel:

$$\lim_{x\to\infty} \frac{\ln(x)^n}{x} = 0$$

für alle $n \in \mathbb{N}$.

Linearformen (Dualraum)

Aufgaben: 5

```
> restart;
> with(LinearAlgebra):
```

Duale Basis

Linearformen tauchen bei der Beschreibung oder approximativen Beschreibung vieler Objekte, z. B. Funktionen in der Analysis, auf. Leben diese Objekte in einem endlich dimensionalen Vektorraum, so ist die Sache übersichtlich, anderenfalls häufig subtil.

MATH: Sei V ein K-Vektorraum. Eine **Linearform** (**lineares Funktional** oder **Kovektor**) auf oder von V ist eine lineare Abbildung von V nach K.

MATH: Ist V ein K-Vektorraum, so heißt der Vektorraum aller Linearformen auf V der **Dualraum** von V und er wird mit

$$V^* := \mathrm{Hom}(V, K)$$

bezeichnet.

WIEDERHOLUNG: Verifiziere die Vektorraumaxiome für V^*.

Ist $\beta \in (V^*)^n$ linear unabhängig, so gibt es für jedes n-Tupel $z \in K^n$ mindestens ein $v \in V$ mit

$$\beta_i(v) = z_i$$

für $i = 1, \ldots, n$.

Klar: v ist eindeutig bis auf Addition eines

$$w \in \bigcap_{i=1}^{n} \mathrm{Kern}(\beta_i).$$

MAPLE u. MATH: Haben wir einen endlich dimensionalen Spaltenraum als Vektorraum V, so bietet es sich an, V^* als Zeilenraum entsprechender Größe zu realisieren: Jede Linearform $\beta \in V^*$ kann mit ihrer Matrix bezüglich einer fest gewählten Basis von V und der Standardbasis (1) von K identifiziert werden:

```
> v:=Vector(5,[1,2,3,4,5]);
```

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \end{bmatrix} \qquad \textbf{(3.1.1)}$$

```
> beta:=Vector[row](5,[2,3,4,5,6]);
```

$$\beta := \begin{bmatrix} 2 & 3 & 4 & 5 & 6 \end{bmatrix} \qquad \textbf{(3.1.2)}$$

```
> beta.v;
```

$$70 \qquad \textbf{(3.1.3)}$$

MATH: Ist $B \in V^n$ eine Basis von V, so heißt $\beta \in (V^*)^n$ die zu B **duale Basis**, falls

$$\beta_i(B_j) = \delta_{ij} := \begin{cases} 1 & i = j \\ 0 & i \neq j \end{cases}$$

(d.h. δ_{ij} ist das Kronecker-Delta).

Man nennt (B, β) auch ein **duales Basispaar**. Offenbar bestimmt β auch B, so wie

β durch B bestimmt ist. Die Idee der Dualbasis ergibt sich aus folgender Darstellung eines beliebigen Vektors $v \in V$:

$$v = \beta_1(v)B_1 + \ldots + \beta_n(v)B_n.$$

DENKANSTOSS: Vergleiche mit den Fourierkoeffizienten im Falle Euklidischer Vektorräume.

ÜBUNG [04]:

$B := (\sin, \cos) \in (\mathbb{R}^{\mathbb{R}})^2$ ist offenbar linear unabhängig, also Basis des von sin und cos erzeugten Teilraumes $V := \langle \sin, \cos \rangle \leq \mathbb{R}^{\mathbb{R}}$.

Sei $(\alpha, \beta) \in (V^*)^2$ mit

$$\alpha : V \rightarrow \mathbb{R} : p \mapsto p\left(\frac{\pi}{2}\right)$$

und

$$\beta : V \rightarrow \mathbb{R} : p \mapsto p(0).$$

Zeige:

a) α und β sind Linearformen auf V, also Elemente von V^*.

b) (α, β) ist die Dualbasis von B.

ÜBUNG [05]:

Sei $V := \mathbb{Q}^{3 \times 1}$ und $b_1 \in V$ sowie $\beta_2, \beta_3 \in V^*$ wie unten gegeben.

a) Finde alle möglichen dualen Basispaare (b, β) von V und V^*, die durch Ergänzung von b_1 und $\beta_2, \beta_3 \in V^*$ zustandekommen.

b) Warum existieren solche Basispaare?

```
> b[1]:=Vector(3,[1,2,3]);
  beta[2]:=Vector[row](3,[3,0,-1]);
  beta[3]:=Vector[row](3,[2,-1,0]);
```

$$b_1 := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$\beta_2 := \begin{bmatrix} 3 & 0 & -1 \end{bmatrix}$$

$$\beta_3 := \begin{bmatrix} 2 & -1 & 0 \end{bmatrix} \quad \textbf{(3.1.4)}$$

Beispiel: Lagrange-Interpolation

MATH:

$$K[x]_{<n} := \{\, p \in K[x] \mid p = 0 \text{ oder } \mathrm{Grad}(p) < n \}$$

ist ein n-dimenisonaler K-Vektorraum. In seinem Dualraum betrachten wir Linearformen

$$\varepsilon_i : K[x]_{<n} \to K : p(x) \mapsto p(a_i),$$

mit n paarweise verschiedenen, fest gewählten $a_i \in K$. Offenbar ist $\varepsilon := (\varepsilon_1, \ldots, \varepsilon_n)$ linear unabhängig, denn für

$$p_i := \prod_{j=1,\, j\neq i}^{n} (x - a_j)$$

erhält man

$$\varepsilon_i(p_j) = 0 \text{ für } i \neq j$$

und

$$\varepsilon_i(p_i) \neq 0.$$

ÜBUNG [06]:

a) Gib eine Basis B von $K[x]_{<n}$ an, sodass ε dual zu B ist.

b) Was hat diese Basis mit einer Interpolation zu tun?
(Hinweis: Benutze B, um für eine beliebige Funktion $f : K \to K$ das eindeutige Polynom $p \in K[x]_{<n}$ anzugeben mit $f(a_i) = p(a_i)$ für $i = 1, \ldots, n$.)

c) Rechne und plotte ein Beispiel einer Interpolation.

MATH: Seien $a_1, \ldots, a_n \in \mathrm{K}$ paarweise verschiedene Punkte. Die Polynome

$$l_j := \frac{\prod_{i=1,\, i\neq j}^{n} (x - a_i)}{\prod_{i=1,\, i\neq j}^{n} (a_j - a_i)} \in K[x]$$

heißen **Lagrange-Fundamentalpolynome**. Beachte: $l_j(a_i) = \delta_{ij}$.

Für eine Funktion $f : \mathrm{K} \to \mathrm{K}$ heißt

$$Lf : \mathrm{K} \to \mathrm{K} : x \mapsto \sum_{i=1}^{n} f(a_i) \cdot l_i(x)$$

die **Lagrange'sche Interpolationsfunktion.**

BEISPIEL: Wir betrachten die Lagrangepolynome in den Punkten:

```
> p:=[1,2,4];
```

$$p := [1, 2, 4] \tag{3.2.1}$$

```
> lp1:=(x-p[2])*(x-p[3])/((p[1]-p[2])*(p[1]-p[3]));
```

$$lp1 := \frac{1}{3}\,(x-2)\,(x-4) \tag{3.2.2}$$

```
> lp2:=(x-p[1])*(x-p[3])/((p[2]-p[1])*(p[2]-p[3]));
```

$$lp2 := -\frac{1}{2}\,(x-1)\,(x-4) \tag{3.2.3}$$

```
> lp3:=(x-p[1])*(x-p[2])/((p[3]-p[1])*(p[3]-p[2]));
```

$$lp3 := \frac{1}{6}\,(x-1)\,(x-2) \tag{3.2.4}$$

```
> plot([lp1,lp2,lp3],x=p[1]-1..p[3]+1,y=-1..1.2,color=[red,
  green,blue]);
```



Mit Hilfe dieser Lagrangepolynome bekommen wir die Lagrange'sche Interpolationsfunktion der Funktion *f*, die in den Punkten 1, 2, 4 mit *f* übereinstimmt.

```
> f:=sin;
```

$$f := \sin \tag{3.2.5}$$

```
> Lf:=collect(lp1*f(p[1])+lp2*f(p[2])+lp3*f(p[3]),x);
```

$$Lf := \left(-\frac{1}{2}\sin(2) + \frac{1}{3}\sin(1) + \frac{1}{6}\sin(4)\right)x^2 + \left(-2\sin(1) + \frac{5}{2}\sin(2) - \frac{1}{2}\sin(4)\right)x + \frac{8}{3}\sin(1) + \frac{1}{3}\sin(4) - 2\sin(2) \tag{3.2.6}$$

```
> plot([f(x),Lf],x=-6..6,y=-3..1.5, color=[red,green]);
```




```
> plot([f(x)-Lf],x=0..5, y=-1..1);
```



Man sieht hier aber auch, dass je nach Wahl (und Anzahl) der Punkte die Approximation unterschiedlich gut ist. (Experimentiere mit verschiedenen Punkten.)

Transponierte Abbildung

MATH: Sind V, W $K-$Vektorräume mit einer K-linearen Abbildung

$$\alpha: V \to W,$$

so ist die **Transponierte** von α, kurz α^{tr}, definiert als

$$\alpha^{tr}: W^* \to V^*: \delta \mapsto \delta \circ \alpha.$$

Also ist das Diagramm

$$\begin{tikzcd} V \arrow[r, "\alpha"] \arrow[dr, "\alpha^{tr}(\delta)"'] & W \arrow[d, "\delta"] \\ & K \end{tikzcd}$$

kommutativ.

Bezieht man V auf die Basis $B \in V^n$ und W auf die Basis $C \in W^m$, so bekommen wir die Matrix

$$A = {}^C\alpha^B$$

zur Beschreibung von α. Bezieht man die Dualräume V^* und W^* auf die jeweiligen Dualbasen $\beta \in (V^*)^n$ und $\gamma \in (W^*)^m$ von B und C, so ist die Matrix von α^{tr} gegeben durch

$$^{\beta}(\alpha^{tr})^{\gamma} = A^{tr}.$$

MATH: Die **formale Ableitung** (oder **formale Differentiation**) für Polynome in $K[x]$ ist definiert als

$$\partial: K[x] \to K[x]: \sum_{i=0}^{n} a_i \cdot x^i \mapsto \sum_{j=0}^{n-1} a_{j+1} \cdot (j+1) \cdot x^j.$$

Dies ist eine lineare Abbildung, die $K[x]_{<n}$ in sich abbildet.

```
> p:=add(a[i]*x^i,i=0..10);
```

$$p := x^{10} a_{10} + x^9 a_9 + x^8 a_8 + x^7 a_7 + x^6 a_6 + x^5 a_5 + x^4 a_4 + x^3 a_3 + x^2 a_2 + x a_1 + a_0 \quad \textbf{(3.3.1)}$$

```
> diff(p,x);
```

$$10 x^9 a_{10} + 9 x^8 a_9 + 8 x^7 a_8 + 7 x^6 a_7 + 6 x^5 a_6 + 5 x^4 a_5 + 4 x^3 a_4 + 3 x^2 a_3 + 2 x a_2 + a_1 \quad \textbf{(3.3.2)}$$

```
> diff(diff(p,x),x);
  diff(p,x$2);
```

$$90 x^8 a_{10} + 72 x^7 a_9 + 56 x^6 a_8 + 42 x^5 a_7 + 30 x^4 a_6 + 20 x^3 a_5 + 12 x^2 a_4 + 6 x a_3 + 2 a_2$$

$$90 x^8 a_{10} + 72 x^7 a_9 + 56 x^6 a_8 + 42 x^5 a_7 + 30 x^4 a_6 + 20 x^3 a_5 + 12 x^2 a_4 \quad \textbf{(3.3.3)}$$

$+6\,x\,a_3+2\,a_2$

Maple hat diese lineare Abbildung also fest eingebaut, sogar ihre Potenzen. Im folgenden setzen wir für unseren Körper K voraus, dass $n \cdot 1 \neq 0 \in K$ für alle $n \in \mathbb{N}$ ist. (Beispiele für solche gutartigen Körper: $\mathbb{Q}$, $\mathbb{R}$, $\mathbb{C}$.)

MATH: Für jedes $k \in K$ ist

$$\varepsilon_k : K[x]_{<n} \to K : p(x) \mapsto p(k)$$

ein Element des Dualraumes $\left(K[x]_{<n}\right)^*$ von $K[x]_{<n}$.

ÜBUNG [07]:

1.) Zeige: Die zu der Basis $\left(1, x, x^2, \dots, x^9\right)$ von $K[x]_{<10}$ duale Basis von $\left(K[x]_{<10}\right)^*$ ist gegeben durch

$$\left(\frac{1}{0!}\,\varepsilon_0, \frac{1}{1!}\varepsilon_0 \circ \partial, \frac{1}{2!}\varepsilon_0 \circ \partial^2, \dots, \frac{1}{9!}\varepsilon_0 \circ \partial^9\right).$$

2.) Bestimme die zu

$$\partial: K[x]_{<n} \to K[x]_{<n}: \sum_{i=0}^{n-1} a_i \cdot x^i \mapsto \sum_{j=0}^{n-2} a_{j+1} \cdot (j+1) \cdot x^j$$

transponierte Abbildung.

3.) Schreibe für $n=6$ die Matrix von ∂ und von ∂^{tr} hin.

Wenn wir also den Entwicklungskoeffizienten von x^{10} von

$$\left(x^2+x+1\right)^{10}$$

in der Basis der x^i haben wollen, können wir sehr schnell rechnen, da

$$a_i = \frac{1}{i!} \cdot \varepsilon_0 \circ \partial^i(\dots) =$$

```
> subs(x=0,diff((x^2+x+1)^10, x$10))/10!;
```

$$8953 \qquad \textbf{(3.3.4)}$$

In der Tat sehen wir:

```
> expand((x^2+x+1)^10);
  coeff(%, x, 10);
```

$$x^{20}+10\,x^{19}+55\,x^{18}+210\,x^{17}+615\,x^{16}+1452\,x^{15}+2850\,x^{14}+4740\,x^{13}+6765\,x^{12}+8350\,x^{11}+8953\,x^{10}+8350\,x^9+6765\,x^8+4740\,x^7+2850\,x^6+1452\,x^5+615\,x^4+210\,x^3+55\,x^2+10\,x+1$$

$$8953 \qquad \textbf{(3.3.5)}$$

ÜBUNG [08]:

Sei V ein K-Vektorraum und $\varphi : V \to K$ linear, also $\varphi \in V^*$. Zeige: Für $\varphi^{tr} : K^* \to V^*$ gilt:

$$\varphi^{tr}(Id_K) = \varphi.$$